

Z. CYLKOWSKI (Wrocław)

APPROXIMATE FUNCTION EXPANSION INTO A CHEBYSHEV SERIES

1. Procedure declaration.

procedure *TChex*(*p*, *q*, *x*, *f*, *n*, *a*, *ex*);

value *p*, *q*;

integer *n*;

real *p*, *q*, *x*, *f*;

Boolean *ex*;

array *a*;

comment For a given natural number $N \geq 2$ the procedure *TChex* calculates the approximate values

$$(1) \quad a_0^{(n)}, a_1^{(n)}, \dots, a_n^{(n)} \quad (n = 2, 4, 8, \dots, n \leq N)$$

of the coefficients of the expansion of $f(x)$ in the interval $\langle p, q \rangle$ into a series of Chebyshev polynomials of first kind, i.e. it calculates numbers (1) such that for $x \in \langle p, q \rangle$ holds

$$(2) \quad f(x) \approx \frac{1}{2} a_0^{(n)} + \sum_{k=1}^n a_k^{(n)} T_k \left(\frac{2x - p - q}{q - p} \right) \quad (T_k(t) = \cos(k \arccos t)).$$

This equality is accurate if $f(x)$ is a polynomial of degree at most n .

Data:

p, *q* — interval limits $\langle p, q \rangle$,

f — arithmetical expression with value $f(x)$ depending upon *x*,

n — natural number N .

Results:

n — upper summation limit in (2), a number of the form 2^l (l — natural),

$a[0:n]$ — array of approximate values of (1) ($a[k] = a_k^{(n)}$ for $k = 0, 1, \dots, n$).

Remark: During calculations the variable n assumes successively the values 2, 4, 8, ... and for each of them the coefficients of (1)

are found. The final value of n is always not greater than N but it may depend upon the value of ex .

Other parameters:

- x — the variable occurring in the 4-th actual parameter,
- ex — Boolean expression which may depend upon the actual values of $n, a[0:n]$. If $ex = \mathbf{true}$ while doubling the value of n then the procedure calculates the coefficients (1) for the new n , if $ex = \mathbf{false}$ the calculations are finished.

Remarks:

1° The coefficients $a_k^{(n/2)}$ ($k = 0, 1, \dots, \frac{1}{2}n$) may be calculated from

$$(3) \quad a_k^{(n/2)} = \begin{cases} a_0^{(n)} + 2a_n^{(n)} & (k = 0), \\ a_k^{(n)} + a_{n-k}^{(n)} & (k = 1, 2, \dots, \frac{1}{2}n - 1), \\ a_{n/2}^{(n)} & (k = \frac{1}{2}n). \end{cases}$$

2° The coefficients (1) with lower indices being only either even or odd may be obtained from the fact that if we have in the interval $\langle -1, 1 \rangle$ the formula

$$(4) \quad g(x) \stackrel{\text{df}}{=} f\left(\frac{1}{2}(p+q) + \frac{1}{2}(q-p)x\right) = \frac{1}{2}a_0 + \sum_{k=1}^n a_k T_k(x),$$

then in the interval $\langle 0, 1 \rangle$ holds

$$(5) \quad g(\sqrt{x}) + g(-\sqrt{x}) = a_0 + 2 \sum_{k=1}^{n/2} a_{2k} T_k(2x-1),$$

$$\sqrt{x}(g(\sqrt{x}) - g(-\sqrt{x})) = a_1 + \sum_{k=1}^{n/2} (a_{2k-1} + a_{2k+1}) T_k(2x-1) \quad (a_{n+1} = 0);$$

begin

integer $h, h2, i0, i1, j, k, m, n1, n2$;

real $a0, a1, a2, c, c2$;

$m := n$;

$n := n2 := h2 := 2$;

$x := p$;

$a2 := f$;

$x := p := .5 \times (p + q)$;

$a1 := f$;

$x := q$;

$a0 := f$;

```

q: = q - p;
c: = .5 × (a0 + a2);
a[0]: = c + a1;
a[1]: = .5 × (a0 - a2);
a[2]: = .5 × (c - a1);
for n2: = n2 + n2 while ex ∧ n2 ≤ m do
  begin
    n1: = h2 + 1;
    a0: = 3.14159265359/h2;
    for j: = h2 step -1 until 1 do
      begin
        c2: = cos(a0 × (j - .5));
        x: = p + q × c2;
        a[h2 + j]: = f
      end j;
    c2: = c2 + c2;
    for h: = h2 while h > 1 do
      begin
        h2: = .5 × h;
        a2: = c: = c2;
        c2: = c2 × c2 - 2;
        for j: = 0, k while j < h2 do
          begin
            k: = j + 1;
            i1: = n1 - k;
            for i0: = n1 + j step h until n2 do
              begin
                i1: = i1 + h;
                a0: = a[i0];
                a1: = a[i1];
                a[i0]: = c × (a0 - a1);
                a[i1]: = a0 + a1;
                c: = -c
              end
            end
          end
        end
      end
    end
  end

```

```

        end i0;
        a0: = a2;
        a2: = abs(c);
        c: = c2 × a2 − a0
    end j
end h;
k: = h: = .25 × n2;
h2: = n1 − 3;
for j: = 1 step 1 until h2 do
begin
    if j < k
    then begin
        i0: = n2 − j;
        i1: = n2 − k;
        a0: = a[i0];
        a[i0]: = a[i1];
        a[i1]: = a0
    end j < k;
    i0: = h;
aa: if i0 ≤ k
    then begin
        k: = k − i0;
        i0: = .5 × i0;
        go to aa
    end i0 ≤ k
    else k: = k + i0
    end j;
for h2: = h while h > 1 do
begin
    h: = .5 × h2;
    i0: = n2 − h;
    for k: = i0 − h + 1 step 1 until i0 do

```

```

begin
   $a0 := a[k];$ 
  for  $j := k - h2$  step  $-h2$  until  $n1$  do
     $a0 := a[j] := a[j] - a0$ 
  end  $k$ 
end  $h2$ ;
 $a2 := 1/n2$ ;
 $a[0] := .5 \times a[0]$ ;
for  $j := n1$  step  $1$  until  $n2$  do
  begin
     $k := n2 - j$ ;
     $a0 := .5 \times a[k]$ ;
     $a1 := a2 \times a[j]$ ;
     $a[k] := a0 + a1$ ;
     $a[j] := a0 - a1$ 
  end  $j$ ;
   $a[0] := 2 \times a[0]$ ;
   $n := h2 := n2$ 
end  $n2$ 
end TChex

```

2. Method used. Let

$$g(x) = f\left(\frac{1}{2}(p+q) + \frac{1}{2}(q-p)x\right) \quad (x \in \langle -1, 1 \rangle),$$

$$x_j^{(n)} = \cos \frac{j\pi}{n} \quad (n = 2, 4, 8, \dots; j = 0, 1, \dots, n),$$

$$\delta_{kn} = \begin{cases} 0 & (k \neq n), \\ 1 & (k = n). \end{cases}$$

Procedure *TChex* uses the formulae given by Clenshaw [1]

$$(6) \quad a_k^{(n)} = \frac{2 - \delta_{kn}}{n} \sum_{j=0}^n {}'' g(x_j^{(n)}) T_k(x_j^{(n)}) \quad (k = 0, 1, \dots, n)$$

(\sum'' denotes that the first and last summands should be halved)

to calculate $a_0^{(2)}, a_1^{(2)}, a_2^{(2)}$. All other $a_k^{(2n)}$ are calculated after formulae which follow from (6):

$$a_k^{(2n)} = \frac{1}{2}(a_k^{(n)} + b_k^{(n)}) \quad (k = 0, 1, \dots, n-1), \quad a_n^{(2n)} = a_n^{(n)},$$

$$a_{2n-k}^{(2n)} = \frac{1}{2}(a_k^{(n)} - b_k^{(n)}) \quad (k = 1, 2, \dots, n-1), \quad a_{2n}^{(2n)} = \frac{1}{2}(a_0^{(n)} - b_0^{(n)}),$$

where $b_k^{(n)}$ are expressed by formulae given by Lanczos [3] (p. XVII):

$$(7) \quad b_k^{(n)} = \frac{2}{n} \sum_{j=1}^n g(x_{2j-1}^{(2n)}) T_k(x_{2j-1}^{(2n)}) \quad (k = 0, 1, \dots, n-1).$$

Since the numbers given by (7) are Chebyshev coefficients of the polynomial P satisfying

$$P(x_{2j-1}^{(2n)}) = g(x_{2j-1}^{(2n)}) \quad (j = 1, 2, \dots, n),$$

degree of $P \leq n$, therefore, from (5), they are calculated from the coefficients of the polynomial

$$P_0(x) = P(\sqrt{\frac{1}{2}x + \frac{1}{2}}) + P(-\sqrt{\frac{1}{2}x + \frac{1}{2}})$$

and

$$P_1(x) = \sqrt{\frac{1}{2}x + \frac{1}{2}} (P(\sqrt{\frac{1}{2}x + \frac{1}{2}}) - P(-\sqrt{\frac{1}{2}x + \frac{1}{2}})).$$

The polynomials P_0 and P_1 are of degree at most $\frac{1}{2}n$ and satisfy the following interpolation conditions:

$$P_0(x_{2j-1}^{(n)}) = P(x_{2j-1}^{(2n)}) + P(x_{2(n-j)+1}^{(2n)}),$$

$$P_1(x_{2j-1}^{(n)}) = x_{2j-1}^{(2n)} (P(x_{2j-1}^{(2n)}) - P(x_{2(n-j)+1}^{(2n)})),$$

where $j = 1, 2, \dots, \frac{1}{2}n$.

If $n = 2$, the coefficients of the polynomials P_0 and P_1 are already obtained (the coefficient of $T_n(x)$ is not needed); otherwise formulae (5) are used repeatedly which leads to n polynomials of degree at most one which satisfy one interpolation condition each.

3. Coefficients of the expansion into a series of Chebyshev polynomials of second kind. If, as before,

$$g(x) = \frac{1}{2}a_0 + \sum_{k=1}^{\infty} a_k T_k(x) \quad (x \in \langle -1, 1 \rangle)$$

and, in addition,

$$g(x) = \sum_{k=0}^{\infty} \alpha_k U_k(x) \quad \left(U_k(x) = \frac{\sin((k+1)\arccos x)}{\sqrt{1-x^2}} \right),$$

then it follows from the known relation $T_k(x) = \frac{1}{2}(U_k(x) - U_{k-2}(x))$ that

$$\alpha_k = \frac{1}{2}(a_k - a_{k+2}) \quad (k = 0, 1, \dots),$$

and Chebyshev coefficients relevant to Chebyshev polynomials of 2-nd kind may easily be obtained by procedure *TChex*.

In addition, the expansion of function g into a series of polynomials $U_k(x)$ may be obtained directly using a method similar to that described in Section 2. Thus, for $x \in \langle 0, 1 \rangle$ the following formulae hold:

$$(8) \quad \begin{aligned} g(\sqrt{x}) + g(-\sqrt{x}) &= 2 \sum_{k=0}^{\infty} (a_{2k} + a_{2k+2}) U_k(2x-1), \\ \frac{g(\sqrt{x}) - g(-\sqrt{x})}{\sqrt{x}} &= 4 \sum_{k=0}^{\infty} a_{2k+1} U_k(2x-1); \end{aligned}$$

they are used for calculating $\beta_k^{(n)}$, given by (10).

The approximate values

$$(9) \quad \alpha_0^{(n)}, \alpha_1^{(n)}, \dots, \alpha_n^{(n)} \quad (n = 2^l - 2; l = 1, 2, \dots)$$

of the Chebyshev coefficients are calculated from the recurrent formulae

$$\begin{aligned} \alpha_0^{(n)} &= g(0), \\ \left. \begin{aligned} \alpha_k^{(2n+2)} &= \frac{1}{2}(\beta_k^{(n)} + \alpha_k^{(n)}) \\ \alpha_{2n+2-k}^{(2n+2)} &= \frac{1}{2}(\beta_k^{(n)} - \alpha_k^{(n)}) \end{aligned} \right\} (k = 0, 1, \dots, n), \\ \alpha_{n+1}^{(2n+2)} &= \frac{1}{2}\beta_{n+1}^{(n)}, \end{aligned}$$

where

$$(10) \quad \beta_k^{(n)} = \frac{2}{n+2} \sum_{j=1}^{n+2} (1 - (x_{2j-1}^{(2n+4)})^2) g(x_{2j-1}^{(2n+4)}) U_k(x_{2j-1}^{(2n+4)})$$

$$(k = 0, 1, \dots, n+1).$$

These formulae follow from an application of a known quadrature (see e.g. [2], p. 111, formula (7.3.7)):

$$\begin{aligned} \alpha_k &= \frac{2}{\pi} \int_{-1}^1 \sqrt{1-x^2} g(x) U_k(x) dx \\ &\approx \frac{2}{n+2} \sum_{j=1}^{n+1} (1 - (x_j^{(n+2)})^2) g(x_j^{(n+2)}) U_k(x_j^{(n+2)}) = \alpha_k^{(n)}. \end{aligned}$$

The numbers $\beta_k^{(n)}$, similarly as $b_k^{(n)}$ before, are obtained by applying (8) to the polynomial

$$Q(x) = \sum_{k=0}^n \beta_k^{(n)} U_k(x) + \frac{1}{2} \beta_{n+1}^{(n)} U_{n+1}(x)$$

and to the polynomials which arise from Q in such a way.

The definition of Q and [2] (p. 111, formula (7.3.6)) yield

$$\begin{aligned} \beta_k^{(n)} &= \frac{2 + 2\delta_{k,n+1}}{\pi} \int_{-1}^1 \frac{(1-x^2)Q(x)U_k(x)}{\sqrt{1-x^2}} dx \\ &= \frac{2}{n+2} \sum_{j=1}^{n+2} (1 - (x_{2j-1}^{(2n+4)})^2) Q(x_{2j-1}^{(2n+4)}) U_k(x_{2j-1}^{(2n+4)}) \quad (k = 0, 1, \dots, n+1), \end{aligned}$$

and, because the interpolation polynomial is uniquely determined, there is

$$Q(x_{2j-1}^{(2n+4)}) = g(x_{2j-1}^{(2n+4)}) \quad (j = 1, 2, \dots, n+2).$$

The following list contains in the left column the lines or groups of consecutive lines which are to be removed from the body of procedure *TChex* and replaced by the lines from the right column in order to obtain a procedure calculating the coefficients (9), i.e. $a[k] = \alpha_k^{(n)}$ ($k = 0, 1, \dots, n$).

integer $h, h2, i0, i1, j, k, m, n1, n2$;	integer $h, h2, i0, i1, j, k, m, n1, n11, n2$;
$n := n2 := h2 := 2$;	$n := 0$;
$x := p$;	$n1 := 1$;
$a2 := f$;	
$a1 := f$;	
$x := q$;	
$a0 := f$;	
$c := .5 \times (a0 + a2)$;	$a[0] := f$;
$a[0] := c + a1$;	for $n2 := n1 + n1$ while
$a[1] := .5 \times (a0 - a2)$;	$ex \wedge n2 \leq m$ do
$a[2] := .5 \times (c - a1)$;	
for $n2 := n2 + n2$ while	
$ex \wedge n2 \leq m$ do	
$n1 := h2 + 1$;	$n11 := h2 := n1 + 1$;
for $j := h2$ step -1 until 1 do	for $j := n1$ step -1 until 0 do
$c2 := \cos(a0 \times (j - .5))$;	$c2 := \cos(a0 \times (j + .5))$;


```

a[h2+j]: = f
a[i0]: = c × (a0 - a1);
k: = h: = .25 × n2;
h2: = n1 - 3;
for j: = 1 step 1 until h2 do
i0: = n2 - h;
for k: = i0 - h + 1 step 1 until i0 do
for j: = k - h2 step -h2 until n1 do
a2: = 1/n2;
a[0]: = .5 × a[0];
for j: = n1 step 1 until n2 do
a[j]: = a0 - a1
a[0]: = 2 × a[0];
n: = h2: = n2

a[n1+j]: = f
a[i0]: = (a0 - a1)/c;
k: = h: = .5 × n11;
for j: = 1 step 1 until n do
i0: = n1 + h2 - 1;
for k: = n1 + h step 1 until i0 do
for j: = k + h2 step h2 until n2 do
a2: = .5/n11;
a[n1]: = a[n1] × (a2 + a2);
for j: = n11 step 1 until n2 do
a[j]: = a1 - a0
n1: = n2 + 1;
n: = n2

```

4. Certification. For the function $f(x) = \log x$ and the interval $\langle p, q \rangle = \langle \frac{1}{2}, \frac{3}{2} \rangle$ the results given in Table 1 were obtained on the ODR A 1204 computer (accuracy: 2^{-37}). To test the influence of rounding errors, the maximum deviations Δ_n of the approximate coefficients (1) from the appropriate accurate Chebyshev coefficients have been also calculated (Table 2).

TABLE 1

k	$a_k^{(2)}$	$a_k^{(4)}$	$a_k^{(8)}$
0	-.1438410362	-.1386862144	-.1386729286
1	.5493061443	.5359283009	.5358983852
2	-.0719205181	-.0719205181	-.0717967711
3		.0133778435	.0128252633
4		-.0025774109	-.0025774109
5			.0005525802
6			-.0001237470
7			.0000299156
8			-.0000066429

TABLE 2

n	Δ_n
2	.0134077595
4	.0005525858
8	.0000015822
16	.0000000000
32	.0000000000
64	.0000000004
128	.0000000004
256	.0000000044
512	.0000000177
1024	.0000000926

If the actual parameters of procedure *TChex* are not involved, the calculation time is nearly proportional to $\nu(\log_2 \nu + 1)$, where ν is the final value of parameter n .

References

- [1] C. W. Clenshaw, *Chebyshev series for mathematical functions*, London 1962.
 [2] В. И. Крылов, *Приближенное вычисление интегралов*, Москва 1959.
 [3] *Tables of Chebyshev polynomials*, Washington 1952.

DEPT. OF NUMERICAL METHODS
 MATHEMATICAL INSTITUTE
 UNIVERSITY OF WROCLAW

Received on 10. 10. 1970

Z. CYLKOWSKI (Wrocław)

ALGORITM 13

PRZYBLIŻONE ROZWIJANIE FUNKCJI W SZEREG CZEBYSZEWA

STRESZCZENIE

Procedura *TChex* dla danej liczby naturalnej $N > 2$ oblicza przybliżone wartości (1) współczynników rozwinięcia funkcji $f(x)$ w przedziale $\langle p, q \rangle$ w szereg względem wielomianów Czebyszewa I rodzaju, tj. takie liczby (1), że dla $x \in \langle p, q \rangle$ zachodzi równość (2). Równość (2) jest dokładna, jeśli $f(x)$ jest wielomianem co najwyżej n -tego stopnia.

Dane:

p, q — końce przedziału $\langle p, q \rangle$,

f — wyrażenie arytmetyczne o wartości $f(x)$, zależne od parametru x ,

n — liczba naturalna N .

Wyniki:

n — górna granica sumowania w (2), liczba postaci 2^l (l — liczba naturalna),
 $a[0:n]$ — tablica przybliżonych wartości (1) ($a[k] = a_k^{(n)}$ dla $k = 0, 1, \dots, n$).

Uwaga: W czasie obliczeń zmienna n przyjmuje kolejno wartości 2, 4, 8, ... i dla każdej z nich znajduje się współczynniki (1). Końcowa wartość tej zmiennej jest zawsze nie większa od N , ale może zależeć od wartości parametru ex .

Inne parametry:

x — zmienna występująca w czwartym parametrze aktualnym,

ex — wyrażenie boolowskie, które może zależeć od aktualnych wartości n i $a[0:n]$.

Jeśli przy podwojeniu wartości n zachodzi $ex = \text{true}$, to procedura oblicza współczynniki (1) dla nowego n , a jeśli $ex = \text{false}$, to obliczenia kończy się.

Uwagi:

1° Współczynniki $a_k^{(n/2)}$ ($k = 0, 1, \dots, \frac{1}{2}n$) można odtworzyć z wzoru (3).

2° Współczynniki (1) o dolnych wskaźnikach tylko parzystych albo tylko nieparzystych można otrzymać, uwzględniając zależność, że jeśli w przedziale $\langle -1, 1 \rangle$ zachodzi wzór (4), to w przedziale $\langle 0, 1 \rangle$ zachodzą wzory (5).

Użyta w procedurze *TChex* metoda jest opisana w § 2. Metodę tę i treść procedury można zmodyfikować (§ 3) tak, żeby otrzymać współczynniki (9) rozwinięcia funkcji $f(x)$ w szereg względem wielomianów Czebyszewa II rodzaju. Obliczenia kontrolne (§ 4) wykonano na maszynie cyfrowej ODRA 1204, przy czym czas obliczeń był proporcjonalny do wielkości $\nu(\log_2 \nu + 1)$, gdzie ν jest końcową wartością parametru n .
